Supplemental material

Use of Space-time Models to Investigate the Stability of Patterns of Disease

Juan Jose Abellan, Sylvia Richardson and Nicky Best



Supplemental Material, Figure 1: Subset of 309 grid squares (gray) and squares selected (red) to 'bump', in the three different scenarios: $\mathcal{M}_{20}$ (top), $\mathcal{M}_{8}$ (bottom left), and $\mathcal{M}_{1}$ (bottom right).



Supplemental Material, Figure 2: Posterior distribution of the standard deviations of the mixture component. Two chains with overdispersed initial values.



Supplemental Material, Figure 3: Time series plots of the simulated 200 000 values (excluding the first 500) for the mixture components in the case study. Two chains with overdispersed initial values.



Supplemental Material, Figure 4: Time series plots of the simulated 200 000 values (excluding the first 500) for the mixture components thinned by keeping every 100th in the case study. Two chains with overdispersed initial values.

Supplemental Material, Appendix 1

This is the model with two mixture components for the variances of the space-time interactions terms as implemented in WinBUGS.

```
model {

# The model

 for (i in 1:N) {
   for (j in 1:T) {
     cm[i,j] ~ dbin(pi[i,j], births[i,j])
     logit(pi[i,j]) <- inter+lambda[i]+xi[j]+nu[i,j]
     OR[i,j] <- exp(lambda[i]+xi[j]+nu[i,j])
   }
 }



 for (i in 1:N){
   ORlambda[i] <- exp(lambda[i])
   prob.lambda[i] <- step(lambda[i])
 }

 for (j in 1:T){
   ORxi[j] <- exp(xi[j])
   prob.xi[j] <- step(xi[j])
 }



# Prior distributions:

# - Intercept:

 alpha ~ dnorm(0, 0.001)
 ORalpha <- exp(inter)

# - Space:
```

```
 for (i in 1:N){
   lambda[i] ~ dnorm(mu[i], tau.lambda)
 }
 mu[1:N] ~ car.normal(adj[],weights[],num[],tau.mu)

# - Time:

 for(j in 1:T){
   xi[j] ~ dnorm(gamma[j], tau.xi)
 }
 gamma[1:T] ~ car.normal(adj.t[],weights.t[],num.t[],tau.gamma)

# - Interaction:

 for(i in 1:N){
   for(j in 1:T){
   nu[i,j] ~ dnorm(0, tau.nu[ind[i,j]])
   ORnu[i,j] <- exp(nu[i,j])
   prob.nu[i,j] <- step(nu[i,j])
   }
 }

# Calculating empirical variance in time

 for(i in 1:N){ sd.nu[i] <- sd(nu[i, 1:T]) }

# Hyperpriors:

 tau.lambda ~ dgamma(0.5, 0.0005)
 tau.xi ~ dgamma(0.5, 0.0005)
 tau.mu ~ dgamma(0.5, 0.0005)
 tau.gamma ~ dgamma(0.5, 0.0005)

 sigma.lambda <- 1/sqrt(tau.lambda)
 sigma.xi <- 1/sqrt(tau.xi)
 sigma.mu <- 1/sqrt(tau.mu)
 sigma.gamma <- 1/sqrt(tau.gamma)
```

```
 for(i in 1:N){
   for(j in 1:T){
   ind[i,j] ~ dcat(P[])
   }
 }

 sigma.nu[1] ~ dnorm(0, 100)I(0.0,)
 kappa ~ dnorm(0, 0.01)I(0.0,)
 sigma.nu[2] <- sigma.nu[1] + kappa

 P[1:2] ~ ddirch(alpha[])

 for(i in 1:2){ tau.nu[i] <- pow(sigma.nu[i], -2) }

 alpha[1] <- 1
 alpha[2] <- 1

# Weights for adjency matrices in space and time, respectively
 for(k in 1:2210){
   weights[k] <- 1
 }
 for(k in 1:30){
   weights.t[k] <- 1
 }
}
```